
SAS syntax and output for Chapter 2: Multiple Regression with Continuous Predictors
Much of the material in this document builds on the SAS syntax and output for Chapter 1. If necessary, refer back to that document for reminders.

Begin by importing the ‘aggression.txt’ data into SAS:
	data ch1data;
 infile 'c:\aggression.dat' dlm='09'x;
 input age BPAQ AISS alcohol BIS NEOc gender NEOo;
run;



Scatterplots of BPAQ and BIS scores against age (Figure 2.1):

	proc sgplot data=ch1data;
 scatter x=age y=BPAQ; 
run;

	proc sgplot data=ch1data;
 scatter x=age y=BIS; 
run;

The 'corr' procedure gives the correlations along with their 95% confidence intervals:
	proc corr data=ch1data pearson fisher;
 var BPAQ BIS age;
run;



[image: ]

Multiple regression with two predictors
In the 'model' statement of 'proc reg' (used to specify linear regression models and estimate them using OLS), one can list any number of predictor variables on the right-hand side.

Here, BPAQ is regressed on both BIS and age simultaneously:

	proc reg data=ch1data plots=none;
model BPAQ = BIS age / clb;
output out=ch1data r=resid h=hatval rstudent=rstudent cookd=cookd p=predict;
run;
quit;



[image: ]
The output above reproduces the regression model parameter estimates and their associated standard errors and t-tests as well as the multiple R2 value and its corresponding F test reported in Chapter 2.
Because standardized regression coefficients are typically not recommended, the 'reg' procedure does not compute them by default. But if you must, they can be obtained using the 'stb' option:

	proc reg data=ch1data plots=none;
model BPAQ = BIS age / stb;
run;
quit;


[image: ]

Simultaneous regression
Specify and estimate the simultaneous regression model, with BPAQ regressed on BIS, age, NEOo, and NEOc:

	proc reg data=ch1data plots=none;
model BPAQ = BIS age NEOo NEOc / clb;
run;
quit;

The results above match the results in Table 2.4.


Hierarchical regression
First estimate the model regressing BPAQ on BIS and age (referred to as Model 1 in Chapter 2) and the model regressing BPAQ on BIS, age, NEOo, and NEOc (referred to as Model 2 in Chapter 2) with separate 'model' statements. Then test whether R2 significantly differs across these two models with the 'test' statement which specifies a null hypothesis that the partial regression slopes of the two added variables both equal zero:
	proc reg data=ch1data edf outest=r2 plots=none;
   	mod1: model BPAQ = BIS age;
   	mod2: model BPAQ = BIS age NEOo NEOc;
	R2deltaTest: test NEOo=0, NEOc=0;
run;
quit;

	Model 1


	Model 2



	



The results above should match values given in Table 2.5, with the last F test corresponding to the test of R2 (i.e., F(2,270) = 0.67, p = .51).

Regression diagnostics
Returning to the model regressing BPAQ on BIS and age, plot the residuals against the BIS predictor, enhacning the plot with a LOWESS curve and a spread around the curve (Figure 2.6):

	proc sgplot data=ch1data;
 scatter x=BIS y=resid; 
 loess x=BIS y=resid / lineattrs=(color=blue) smooth=0.9 nomarkers Name="Lowess" CLM;
run;

Enhanced scatterplot of residuals by age (Figure 2.7):
	proc sgplot data=ch1data;
 scatter x=age y=resid; 
 loess x=age y=resid / lineattrs=(color=blue) smooth=0.9 nomarkers Name="Lowess" CLM;
run;

Can also plot residuals against fitted values, i.e., predicted values (Figure 2.8):

	proc sgplot data=ch1data;
 scatter x=predict y=resid; 
 loess x=predict y=resid / lineattrs=(color=blue) smooth=0.9 nomarkers Name="Lowess" CLM;
run;

As in Chapter 1, also can plot univariate distribution of residuals (Figure 2.9) using the 'sgplot' procedure:

	proc sgplot data=ch1data;
 histogram resid;
 density resid/lineattrs=(color=black pattern=2);
 density resid/type=kernel lineattrs=(color=black); 
run;

Polynomial regression

To specify and estimate a polynomial regression model, the (potentially) nonlinear predictor can be squared or cubed in 'data step' before being used as new predictor variables within the 'model' statement of ‘proc reg’. As with hierarchical regression, the ‘test’ statement is used to test whether the polynomal terms significantly increase R2:

	data ch1data;
 set ch1data;
  age2 = age**2;
  age3 = age**3;
run;

proc reg data=ch1data edf outest=r2 plots=none;
 mod1: model BPAQ = BIS age;
 mod2: model BPAQ = BIS age age2 age3;
 R2deltaTest: test age2=0, age3=0;
run;
quit;

	



The results from this F-test should match that given in Chapter 2 for the polynomial regression example.

As in Chapter 1, make index plots of hat values, Studentized residuals, and Cook's distance (Figures 2.10 to 2.12):

	data ch1data;
 set ch1data;
 id=_n_;
run;

proc sgplot data=ch1data;
 scatter x=id y=hatval; 
run;

	proc sgplot data=ch1data;
 scatter x=id y=rstudent; 
run;


[image: C:\Users\Ryuen\Dropbox\DaveProject - book -- software syntax\sasG\8.png]

	proc sgplot data=ch1data;
 scatter x=id y=cookd; 
run;

Test R2deltaTest Results for Dependent Variable BPAQ

Test R2deltaTest Results for Dependent Variable BPAQ

Pearson Correlation Statistics (Fisher's  Transformation)

